classical response model: SHG

A medium that has a linear reponse has a quadratic form of the potential.
examples: $a(x^2 + y^2)$, $(a_x x^2 + a_y y^2)$, $(a_x x^2 + a_y y^2 + a_{xy} x y)$
The second two cases would correspond to a birefringent medium.

We need to add a higher-order term to the potential to get a NL response. Start with a potential that has a cross-term:
$(a_x x^2 + a_y y^2 + a_{xy} x y^2)$
This might look like the plot below:

```
ContourPlot[x^2 + y^2 - x y^2, {x, -1, 1}, {y, -1, 1}, Contours → 20]
```



If you look along the x or y axes, the potential is still harmonic. To see a nonlinear reponse in this potential, you have to look off-axis.

```
Plot[x^2 + y^2 - x y^2 /. y → x, {x, -1, 1}]
```



The equations of motion for a bound electron in this medium would look like:
$\boldsymbol{F} = -\nabla(a_x x^2 + a_y y^2 + a_{xy} x y^2)$
without a driving force, we have:
$m \ddot{x} = -2 \gamma \dot{x} - 2 a_x x - a_{xy} y^2$ where we are adding a damping term
$\ddot{x} + 2 \gamma \dot{x} + 2 a_x x + a_{xy} y^2 = 0$
$\ddot{y} + 2 \gamma \dot{y} + 2 a_y x + 2 a_{xy} x y = 0$
Add a driving field. As noted above, we won't get cross terms (i.e. NL source in a direction orthogonal to the driving field) unless the driving field has both x and y components. We will stick to linearly polarized fields for simplicity and because the medium is birefringent , so elliptically polarized light wouldn't stay so.

$$\ddot{x} + 2\,\gamma\,\dot{x} + 2\,a_x\,x + a_{xy}\,y^2 = -\frac{e}{m}\,E_x(t)$$
$$\ddot{y} + 2\,\gamma\,\dot{y} + 2\,a_y\,y + 2\,a_{xy}\,x\,y = -\frac{e}{m}\,E_y(t)$$

Apply perturbative approach to the solution:
$x(t) = \lambda\,x^{(1)} + \lambda^2\,x^{(2)} + \lambda^3\,x^{(3)}\ldots$
$y(t) = \lambda\,y^{(1)} + \lambda^2\,y^{(2)} + \lambda^3\,y^{(3)}\ldots$
and treat the driving field as a first-order

Lowest order equation:
$$\ddot{x}^{(1)} + 2\,\gamma\,\dot{x}^{(1)} + 2\,a_x\,x^{(1)} = -\frac{e}{m}\,E_{x0}\,e^{i\,\omega_0\,t}$$
$$\ddot{y}^{(1)} + 2\,\gamma\,\dot{y}^{(1)} + 2\,a_y\,y^{(1)} = -\frac{e}{m}\,E_{y0}\,e^{i\,\omega_0\,t}$$
next order:
$$\ddot{x}^{(2)} + 2\,\gamma\,\dot{x}^{(2)} + 2\,a_x\,x^{(2)} + a_{xy}\left(y^{(1)}\right)^2 = 0$$
$$\ddot{y}^{(2)} + 2\,\gamma\,\dot{y}^{(2)} + 2\,a_y\,y^{(2)} + 2\,a_{xy}\,x^{(1)}\,y^{(1)} = 0$$
So it seems that if we drive hard in only the y direction, we can generate SH in the x direction. Looking at the potential plot, you can see the slope in the contours along the y-axis.

```
ω0 = 6;
ω = 1;
Tlaser = 2 π / ω;
γ = 0.1;
c = .5;
tMax = 100;
axy = -0.01;
```

```
solution = NDSolve[{
    x''[t] + 2 γ x'[t] + ω0^2 x[t] + axy y[t]^2 == 0,
    y''[t] + 2 γ y'[t] + ω0^2 y[t] + 2 × 0 axy x[t] y[t] == -c Cos[ω t],
    x[0] == 0, x′[0] == 0,
    y[0] == 0, y′[0] == 0}, {x, y}, {t, 0, tMax}, MaxSteps → 10^6]
xNL[t_] = x[t] /.solution[[1, 1]];
yNL[t_] = y[t] /.solution[[1, 2]];
```

```
{{x → InterpolatingFunction[ Domain: {{0., 100.}} Output scalar ],
  y → InterpolatingFunction[ Domain: {{0., 100.}} Output scalar ]}}
```

```
Plot[xNL[t], {t, 0, tMax}]
Plot[yNL[t], {t, 0, tMax}]
```






```
ParametricPlot[{yNL[t], xNL[t]}, {t, 60, 70}, AspectRatio → .1]
```